2.6.12 Messages

Many built-in *Mathematica* functions print messages to warn you of possible errors. Each message that *Mathematica* prints has a unique name, of the form *symbol*::*tag*.

Section 1.3.7 discussed how you can use `On[`*s*`::`*t*`]` and `Off[`*s*`::`*t*`]` to switch on and off the printing of particular messages.

The text of a message with name *s*::*t* is stored as the value of the object *s*::*t*, associated with the symbol *s*. You can change the text for a message simply by redefining the value of *s*::*t*. Messages are usually stored as strings containing `` `i` `` markers, in the form used by `StringForm`.

If you give `Inverse` a singular matrix, it prints a warning message.

```
In[1]:= Inverse[{{1,1},{2,2}}]

Inverse::sing:
   Encountered singular matrix {{1, 1}, {2, 2}}.

Out[1]= Inverse[{{1, 1}, {2, 2}}]
```

The message is stored as a string to be used with `StringForm`.

```
In[2]:= Inverse::sing

Out[2]= Encountered singular matrix `1`.
```

Most messages are associated directly with the functions that generate them. There are, however, some "general" messages, which can be produced by a variety of functions.

If you give the wrong number of arguments to a function *F*, *Mathematica* will warn you by printing the message *F*`::argct`. If *Mathematica* cannot find a message named *F*`::argct`, it will use the text of the "general" message `General::argct` instead. You can use `Off[`*F*`::argct]` to switch off the argument count message specifically for the function *F*. You can also use `Off[General::argct]` to switch off all messages that use the text of the general message.

Mathematica prints a message if you give the wrong number of arguments to a built-in function.

```
In[3]:= Sqrt[a, b]

Sqrt::argct: Sqrt called with 2 arguments.

Out[3]= Sqrt[a, b]
```

The argument count message is a general one, used by many different functions.

```
In[4]:= General::argct

Out[4]= `1` called with `3` arguments.
```

If something goes very wrong with a calculation you are doing, it is common to find that the same warning message is generated over and over again. This is usually more confusing than useful. As a result, *Mathematica* keeps track of all messages that are produced during a particular calculation, and stops printing a particular message if it comes up more than three times. Whenever this happens, *Mathematica* prints the message `General::stop` to let you know. If you really want to see all the messages that *Mathematica* tries to print, you can do this by switching off `General::stop`.

When you write programs in *Mathematica* you will often want to use the same mechanism for messages as built-in *Mathematica* functions do. One reason for this is that functions like `Check` which test for error conditions look for what messages have been produced.

`s::tag = string`	define a message
`Message[s::tag]`	print a message
`Message[s::tag, expr1, ...]`	print a message, replacing successive `` `` `` in it with successive $expr_i$
`Messages[s]`	show the messages associated with s
`Off[s::tag]`	switch off a message, so that it is not printed
`On[s::tag]`	switch on a message

Functions for manipulating messages.

This defines a message, associated with `f`.

```
In[5]:= f::overflow = "Factorial too large."
Out[5]= Factorial too large.
```

If `x > 10`, `f` gives the `overflow` message, then returns `Infinity`.

```
In[6]:= f[x_Integer] :=
        If[x > 10, Message[f::overflow]; Infinity, x!]
```

The `overflow` message is displayed.

```
In[7]:= f[20]
f::overflow: Factorial too large.
Out[7]= Infinity
```

This switches off the `overflow` message.

```
In[8]:= Off[f::overflow]
```

The message is no longer given.

```
In[9]:= f[20]
Out[9]= Infinity
```

This switches on the message again.

```
In[10]:= On[f::overflow]
```

Web sample page from The *Mathematica* Book, First Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-19334-5; softcover ISBN 0-201-19330-2). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.

Now the message is given.

```
In[11]:= f[20]
f::overflow: Factorial too large.
Out[11]= Infinity
```

This shows the messages associated with f.

```
In[12]:= Messages[f]
Out[12]= f::overflow -> Factorial too large.
```

Here is a new message.

```
In[13]:= f::new = "Overflow with x=``, Log[x]=``."
Out[13]= Overflow with x=``, Log[x]=``.
```

This prints the message, with each `` replaced by an expression.

```
In[14]:= Message[f::new, 30, N[Log[30]]]
f::new: Overflow with x=30, Log[x]=3.4012.
```

Web sample page from The *Mathematica* Book, First Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-19334-5; softcover ISBN 0-201-19330-2). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.