2.4 Advanced Topic: Operators

2.4.1 Manipulating the Heads of Expressions

The head *f* of an expression like *f*[x] is itself an expression, and you can manipulate it in a number of ways.

You can use Head to extract the head of any expression.

```
In[1]:= Head[a + b]
Out[1]= Plus
```

You can test whether an expression has a particular head.

```
In[2]:= Head[a + b] == Plus
Out[2]= True
```

You can use /. to make replacements for the heads of expressions.

```
In[3]:= f[1 - x]  /. f -> g
Out[3]= g[1 - x]
```

If you set p1 to p2, then p1 will automatically be replaced by p2. The replacement is done even if p1 appears as a head.

```
In[4]:= p1 = p2;  p1[x, y]
Out[4]= p2[x, y]
```

You can set up patterns in which the head of an expression can be arbitrary.

This defines a value for ffun[*anyhead*[*anyelem*]].

```
In[5]:= ffun[f_[x_]] := r[x, x]
```

The pattern matches this expression.

```
In[6]:= ffun[ h[5] ]
Out[6]= r[5, 5]
```

It also matches this expression, which involves a built-in function.

```
In[7]:= ffun[ Log[w] ]
Out[7]= r[w, w]
```

You can also give as an argument to a function an object which is to be used as the head of another expression.

This definition takes f as an argument, then uses it as the head of an expression.

```
In[8]:= pf[f_, x_] := f[x] + f[1-x]
```

Here f is the “function” hp.

```
In[9]:= pf[hp, 3]
Out[9]= hp[-2] + hp[3]
```

Web sample page from The *Mathematica* Book, First Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-19334-5; softcover ISBN 0-201-19330-2). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.

f can just as well be a built-in function.

```
In[10]:= pf[Log, q]
Out[10]= Log[q] + Log[1 - q]
```

You can use transformation rules to pull out heads of expressions, and then potentially to use them elsewhere.

This pulls out the head of an expression with a single element, and puts it in a list.

```
In[11]:= pull[f_[x_]] := {f, x}
```

This pulls out the head Log.

```
In[12]:= pull[Log[1 + w]]
Out[12]= {Log, 1 + w}
```

This takes any function f[x] as an argument.

```
In[13]:= sym[f_[x_]] := f[x] + f[1 - x]
```

Here f matches Log.

```
In[14]:= sym[Log[w]]
Out[14]= Log[w] + Log[1 - w]
```

Web sample page from The *Mathematica* Book, First Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-19334-5; softcover ISBN 0-201-19330-2). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.