A.1.2 Clebsch-Gordan Coefficients

Clebsch-Gordan coefficients arise in various group-theoretical calculations, particularly those involving addition of angular momenta in quantum mechanics.

```
BeginPackage["Clebsch`"]

Clebsch::usage =
        "Clebsch[{j1,m1},{j2,m2},{j,m}] gives the Clebsch-Gordan coefficient
        for the addition of angular momenta."

Begin["Clebsch`private`"]

C1[j_,j1_,j2_,m1_,m2_] :=
        Block[{k},
             Sum[ (-1)^k
             (k! (j1+j2-j-k)! (j1-m1-k)! (j2+m2-k)! (j-j2+m1+k)! (j-j1-m2+k)!)^-1,
                  {k, Max[0,j2-j-m1,j1-j+m2], Min[j1+j2-j,j1-m1,j2+m2]} ]
        ]

C2[j_,j1_,j2_] := (-j+j1+j2)! (j-j1+j2)! (j+j1-j2)! (1+2j) / (1+j+j1+j2)!

C3[j_,j1_,j2_,m_,m1_,m2_] := (j-m)! (j+m)! (j1-m1)! (j1+m1)! (j2-m2)! (j2+m2)!

Clebsch[ {j1_, m1_}, {j2_, m2_}, {j_, m_} ] :=
                If[ m!=m1+m2 , 0 ,
                        Block[ {t},
                                t = C1[j,j1,j2,m1,m2] ;
                                Sign[t] Sqrt[t^2 C2[j,j1,j2] C3[j,j1,j2,m,m1,m2]]
                        ]
                ]

End[ ]

EndPackage[ ]
```

Definitions of Clebsch-Gordan coefficients.

Notice the use of a private context inside the main `Clebsch`` context, for defining subsidiary functions used by the main `Clebsch` function.

Read in the Clebsch-Gordan coefficient package.

```
In[1]:= <<Clebsch.m
```

This evaluates a particular Clebsch-Gordan coefficient.

```
In[2]:= Clebsch[{2, 0}, {2, 0}, {2, 0}]

              Sqrt[2]
Out[2]= -(----------)
              Sqrt[7]
```

Web sample page from The *Mathematica* Book, First Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-19334-5; softcover ISBN 0-201-19330-2). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.