■ 3.7.7 Basic Matrix Operations

`Transpose[`m`]`	transpose
`Inverse[`m`]`	matrix inverse
`Det[`m`]`	determinant
`Minors[`m`, `k`]`	a list of the $k \times k$ minors of m
`Sum[`m`[[`i`, `i`]], {`i`, Length[`m`]}]`	trace

Some basic matrix operations.

Transposing a matrix interchanges the rows and columns in the matrix. If you transpose an $m \times n$ matrix, you get an $n \times m$ matrix as the result.

Transposing a 2×3 matrix gives a 3×2 result.

```
In[1]:= Transpose[ {{a, b, c}, {ap, bp, cp}} ]
Out[1]= {{a, ap}, {b, bp}, {c, cp}}
```

`Det[`m`]` gives the determinant of a square matrix m. `Minors[`m`, `k`]` gives a list of the determinants of all the $k \times k$ submatrices of m. You can apply `Minors` to rectangular, as well as square, matrices.

Here is the determinant of a simple 2×2 matrix.

```
In[2]:= Det[ {{a, b}, {c, d}} ]
Out[2]= -(b c) + a d
```

This generates a 3×3 matrix, whose i, j^{th} entry is `a[`i`, `j`]`.

```
In[3]:= m = Array[a, {3, 3}]
Out[3]= {{a[1, 1], a[1, 2], a[1, 3]},
     {a[2, 1], a[2, 2], a[2, 3]},
     {a[3, 1], a[3, 2], a[3, 3]}}
```

Here is the determinant of `m`.

```
In[4]:= Det[ m ]
Out[4]= -(a[1, 3] a[2, 2] a[3, 1]) +
     a[1, 2] a[2, 3] a[3, 1] + a[1, 3] a[2, 1] a[3, 2] -
     a[1, 1] a[2, 3] a[3, 2] - a[1, 2] a[2, 1] a[3, 3] +
     a[1, 1] a[2, 2] a[3, 3]
```

Web sample page from The *Mathematica* Book, First Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-19334-5; softcover ISBN 0-201-19330-2). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.

This gives the list of all 2×2 minors of `m`.

```
In[5]:= Minors[m, 2]

Out[5]= {{-(a[1, 2] a[2, 1]) + a[1, 1] a[2, 2],
          -(a[1, 3] a[2, 1]) + a[1, 1] a[2, 3],
          -(a[1, 3] a[2, 2]) + a[1, 2] a[2, 3]},
         {-(a[1, 2] a[3, 1]) + a[1, 1] a[3, 2],
          -(a[1, 3] a[3, 1]) + a[1, 1] a[3, 3],
          -(a[1, 3] a[3, 2]) + a[1, 2] a[3, 3]},
         {-(a[2, 2] a[3, 1]) + a[2, 1] a[3, 2],
          -(a[2, 3] a[3, 1]) + a[2, 1] a[3, 3],
          -(a[2, 3] a[3, 2]) + a[2, 2] a[3, 3]}}
```

You can use `Det` to find the characteristic polynomial for a matrix. Section 3.7.10 discusses ways to find eigenvalues and eigenvectors directly.

Here is a 3×3 matrix.

```
In[6]:= m = Table[ 1/(i + j), {i, 3}, {j, 3} ]
```

$$\mathrm{Out[6]=}\ \{\{\frac{1}{2}, \frac{1}{3}, \frac{1}{4}\}, \{\frac{1}{3}, \frac{1}{4}, \frac{1}{5}\}, \{\frac{1}{4}, \frac{1}{5}, \frac{1}{6}\}\}$$

Following precisely the standard mathematical definition, this gives the characteristic polynomial for `m`.

```
In[7]:= Det[ m - x IdentityMatrix[3] ]
```

$$\mathrm{Out[7]=}\ \frac{1}{43200} - \frac{131\ x}{3600} + \frac{11\ x^2}{12} - x^3$$

There are many other operations on matrices that can be built up from standard *Mathematica* functions. One example is the *trace* or *spur* of a matrix, given by the sum of the terms on the leading diagonal.

Here is a simple 2×2 matrix.

```
In[8]:= m = {{a, b}, {c, d}}

Out[8]= {{a, b}, {c, d}}
```

You can get the trace of the matrix by explicitly constructing a sum of the elements on its leading diagonal.

```
In[9]:= Sum[ m[[i, i]], {i, 2} ]

Out[9]= a + d
```

Web sample page from The *Mathematica* Book, First Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-19334-5; softcover ISBN 0-201-19330-2). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.