■ 2.4.2 Any Expression Can Be a Head

In most cases, you want the head f of an expression like f[x] to be a single symbol. There are, however, some situations where it is convenient to use more complicated expressions as heads.

Mathematica allows you to use any expression as a head.

Here a+b is the head of the expression.

```
In[1]:= (a + b)[x]
Out[1]= (a + b)[x]
```

This has f[p] as a head.

```
In[2]:= f[p][x]
Out[2]= f[p][x]
```

You can use this kind of expression to give "indexed functions" as heads.

```
In[3]:= f[3][x, y]
Out[3]= f[3][x, y]
```

When you give complicated expressions as heads, you must make sure to put in the appropriate parentheses.

```
In[4]:= ((1 + a)(1 + b))[x]
Out[4]= ((1 + a) (1 + b))[x]
```

One situation in which we have already encountered the use of complicated expressions as heads is in working with pure functions (see Section 2.1.12). If you give Function[*args*, *body*] as the head of an expression, then the function is applied to any arguments you supply.

With the head Function[x, x^2], the value of the expression is the square of the argument.

```
In[5]:= Function[x, x^2] [a + b]
                2
Out[5]= (a + b)
```

Another important use of more complicated expressions as heads is in implementing *functionals* and *functional operators* in mathematics.

As one example, consider the operation of differentiation. As will be discussed in Section 3.5.4, an expression like f' represents a *derivative function*, obtained from f by applying a functional operator to it. In *Mathematica*, f' is represented as Derivative[1][f]: the "functional operator" Derivative[1] is applied to f to give another function, represented as f'.

This expression has a head which represents the application of the "functional operator" Derivative[1] to the "function" f.

```
In[6]:= f'[x] // FullForm
Out[6]//FullForm= Derivative[1][f][x]
```

You can replace the head f' by another head, say fp. This effectively takes fp to be a "derivative function" obtained from f.

```
In[7]:= % /. f' -> fp
Out[7]= fp[x]
```