3.2.5 Combinatorial Functions

`n!`	factorial $n(n-1)(n-2) \times \ldots \times 1$
`n!!`	double factorial $n(n-2)(n-4) \times \ldots$
`Binomial[n, m]`	binomial coefficient $\binom{n}{m} = \frac{n!}{m!(n-m)!}$
`Multinomial[n1, n2, ...]`	multinomial coefficient $\frac{(n_1+n_2+\ldots)!}{n_1!n_2!\ldots}$
`BernoulliB[n]`	Bernoulli number B_n
`BernoulliB[n, x]`	Bernoulli polynomial $B_n(x)$
`EulerE[n]`	Euler number E_n
`EulerE[n, x]`	Euler polynomial $E_n(x)$
`StirlingS1[n, m]`	Stirling number of the first kind $S_n^{(m)}$
`StirlingS2[n, m]`	Stirling number of the second kind $\mathcal{S}_n^{(m)}$
`PartitionsP[n]`	the number $p(n)$ of unrestricted partitions of the integer n
`PartitionsQ[n]`	the number $q(n)$ of partitions of n into distinct parts

Combinatorial functions.

The **factorial function** `n!` gives the number of ways of ordering n objects. For non-integer n, the numerical value of $n!$ is obtained from the gamma function, discussed in Section 3.2.10.

The **binomial coefficient** `Binomial[n, m]` can be written as $\binom{n}{m} = \frac{n!}{m!(n-m)!}$. It gives the number of ways of choosing m objects from a collection of n objects, without regard to order. The **Catalan numbers**, which appear in various tree enumeration problems, are given in terms of binomial coefficients $c_n = \frac{1}{n+1}\binom{2n}{n}$.

The **multinomial coefficient** `Multinomial[n1, n2, ...]`, denoted $(N; n_1, n_2, \ldots, n_m) = \frac{N!}{n_1!n_2!\ldots n_m!}$, gives the number of ways of partitioning N distinct objects into m sets, each of size n_i (with $N = \sum_{i=1}^{m} n_i$).

Mathematica gives the exact integer result for the factorial of an integer.

```
In[1]:= 30!
Out[1]= 265252859812191058636308480000000
```

Web sample page from The *Mathematica* Book, First Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-19334-5; softcover ISBN 0-201-19330-2). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.

For non-integers, *Mathematica* evaluates factorials using the gamma function.

```
In[2]:= 3.6!
Out[2]= 13.3813
```

Mathematica can give symbolic results for some binomial coefficients.

```
In[3]:= Binomial[n, 2]
        n (-1 + n)
Out[3]= ----------
            2
```

This gives the number of ways of partitioning $6 + 5 = 11$ objects into sets containing 6 and 5 objects.

```
In[4]:= Multinomial[6, 5]
Out[4]= 462
```

The result is the same as $\binom{11}{6}$.

```
In[5]:= Binomial[11, 6]
Out[5]= 462
```

The **Bernoulli polynomials** `BernoulliB[`n`, `x`]` satisfy the generating function relation $\frac{te^{xt}}{e^t-1} = \sum_{n=0}^{\infty} B_n(x)\frac{t^n}{n!}$. The **Bernoulli numbers** `BernouilliB[`n`]` are given by $B_n = B_n(0)$. The B_n appear as the coefficients of the terms in the Euler-Maclaurin summation formula for approximating integrals.

Numerical values for Bernoulli numbers are needed in many numerical algorithms. You can always get these numerical values by first finding exact rational results using `BernoulliB[`n`]`, and then applying `N`. The function `NBernoulliB[`n`]` gives the numerical value of B_n directly. `NBernoulliB[`n`, `p`]` gives the value to p digit precision.

The **Euler polynomials** `EulerE[`n`, `x`]` have generating function $\frac{2e^{xt}}{e^t+1} = \sum_{n=0}^{\infty} E_n(x)\frac{t^n}{n!}$, and the **Euler numbers** `EulerE[`n`]` are given by $E_n = 2^n E_n(\frac{1}{2})$. The Euler numbers are related to the **Genocchi numbers** by $\mathrm{G}_n = 2^{2-2n} n \mathrm{E}_{2n-1}$.

This gives the second Bernoulli polynomial $B_2(x)$.

```
In[6]:= BernoulliB[2, x]
        1           2
Out[6]= - - x + x
        6
```

You can also get Bernoulli polynomials by explicitly computing the power series for the generating function.

```
In[7]:= Series[t Exp[x t]/(Exp[t] - 1), {t, 0, 4}]
                                                2
                 1                 1    x    x     2
Out[7]= 1 + (-(-) + x) t + (-- - - + --) t  +
                 2                 12   2    2

               2    3
          x   x    x    3       4
         (-- - -- + --) t  + O[t]
          12   4    6
```

Web sample page from The *Mathematica* Book, First Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-19334-5; softcover ISBN 0-201-19330-2). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.

BernoulliB[n] gives exact rational number results for Bernoulli numbers.

```
In[8]:= BernoulliB[20]

          174611
Out[8]= -(------)
           330
```

NBernoulliB[n] gives numerical values for the B_n directly.

```
In[9]:= NBernoulliB[20]

Out[9]= -529.124242424242
```

Stirling numbers show up in many combinatorial enumeration problems. For **Stirling numbers of the first kind** `StirlingS1[`n`, `m`]`, $(-1)^{n-m} S_n^{(m)}$ gives the number of permutations of n elements which contain exactly m cycles. These Stirling numbers satisfy the generating function relation $x(x-1)...(x-n+1) = \sum_{m=0}^{n} S_n^{(m)} x^m$.

Stirling numbers of the second kind `StirlingS2[`n`, `m`]` give the number of ways of partitioning a set of n elements into m non-empty subsets. They satisfy the relation $x^n = \sum_{m=0}^{n} \mathcal{S}_n^{(m)} x(x-1)...(x-m+1)$.

The **partition function** `PartitionsP[`n`]` gives the number of ways of writing the integer n as a sum of positive integers, without regard to order. `PartitionsQ[`n`]` gives the number of ways of writing n as a sum of positive integers, with the constraint that all the integers in each sum are distinct.

This gives a table of Stirling numbers of the first kind.

```
In[10]:= Table[StirlingS1[5, i], {i, 5}]

Out[10]= {24, -50, 35, -10, 1}
```

The Stirling numbers appear as coefficients in this product.

```
In[11]:= Expand[Product[x - i, {i, 0, 4}]]

                     2       3       4    5
Out[11]= 24 x - 50 x  + 35 x  - 10 x  + x
```

This gives the number of partitions of 100, with and without the constraint that the terms should be distinct.

```
In[12]:= {PartitionsQ[100], PartitionsP[100]}

Out[12]= {444793, 190569292}
```

The partition function $p(n)$ increases asymptotically like $e^{\sqrt{n}}$. Note that you cannot simply use `Plot` to generate a plot of a function like `PartitionsP` because the function can only be evaluated with integer arguments.

```
In[13]:= ListPlot[ Table[
         N[Log[ PartitionsP[n] ]], {n, 100} ] ]
```



The functions in this section allow you to *enumerate* various kinds of combinatorial objects. Functions like `Permutations`, discussed in Section 1.7.13, allow you instead to *generate* lists of various combinations of elements.