■ 3.7.6 Matrix Inversion

`Inverse[m]`	find the inverse of a square matrix

Matrix inversion.

Here is a simple 2×2 matrix.

```
In[1]:= m = {{a, b}, {c, d}}
Out[1]= {{a, b}, {c, d}}
```

This gives the inverse of `m`. In producing this formula, *Mathematica* implicitly assumes that the determinant `a d - b c` is nonzero.

```
In[2]:= Inverse[ m ]
                 d                   b
Out[2]= {{-------------, -(-------------)},
          -(b c) + a d     -(b c) + a d

              c                   a
         {-(-------------), -------------}}
            -(b c) + a d    -(b c) + a d
```

Multiplying the inverse by the original matrix should give the identity matrix.

```
In[3]:= % . m
                   b c              a d
Out[3]= {{-(-------------) + -------------, 0},
             -(b c) + a d    -(b c) + a d

                  b c              a d
         {0, -(-------------) + -------------}}
               -(b c) + a d    -(b c) + a d
```

You have to use `Together` to clear the denominators, and get back a standard identity matrix.

```
In[4]:= Together[ % ]
Out[4]= {{1, 0}, {0, 1}}
```

Here is a matrix of rational numbers.

```
In[5]:= hb = Table[1/(i + j), {i, 4}, {j, 4}]
          1  1  1  1     1  1  1  1     1  1  1  1
Out[5]= {{-, -, -, -}, {-, -, -, -}, {-, -, -, -},
          2  3  4  5     3  4  5  6     4  5  6  7

          1  1  1  1
         {-, -, -, -}}
          5  6  7  8
```

Mathematica finds the exact inverse of the matrix.

```
In[6]:= Inverse[hb]
Out[6]= {{200, -1200, 2100, -1120},
         {-1200, 8100, -15120, 8400},
         {2100, -15120, 29400, -16800},
         {-1120, 8400, -16800, 9800}}
```

Web sample page from The *Mathematica* Book, First Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-19334-5; softcover ISBN 0-201-19330-2). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.

Multiplying by the original matrix gives the identity matrix.

```
In[7]:= % . hb
Out[7]= {{1, 0, 0, 0}, {0, 1, 0, 0}, {0, 0, 1, 0},
    {0, 0, 0, 1}}
```

If you try to invert a singular matrix, *Mathematica* prints a warning message, and returns the inverse undone.

```
In[8]:= Inverse[ {{1, 2}, {1, 2}} ]
Inverse::sing:
   Encountered singular matrix {{1, 2}, {1, 2}}.
Out[8]= Inverse[{{1, 2}, {1, 2}}]
```

If you give a matrix with exact symbolic or numerical entries, *Mathematica* gives the exact inverse. If, on the other hand, some of the entries in your matrix are approximate real numbers, then *Mathematica* finds an approximate numerical result.

Here is a matrix containing approximate real numbers.

```
In[9]:= m = {{1.2, 5.7}, {4.2, 5.6}}
Out[9]= {{1.2, 5.7}, {4.2, 5.6}}
```

This finds the numerical inverse.

```
In[10]:= Inverse[ % ]
Out[10]= {{-0.325203, 0.33101}, {0.243902, -0.0696864}}
```

Multiplying by the original matrix gives you an identity matrix with small numerical errors.

```
In[11]:= % . m
                           -16                  -16
Out[11]= {{1., 2.22045 10   }, {-1.11022 10    , 1.}}
```

You can get rid of the small off-diagonal terms using `Chop`.

```
In[12]:= Chop[ % ]
Out[12]= {{1., 0}, {0, 1.}}
```

When you try to invert a matrix with exact entries, *Mathematica* can always tell whether or not the matrix is singular. When you invert a numerical matrix, *Mathematica* can never tell for certain whether or not the matrix is singular: all it can tell is for example that the determinant is small compared to the entries of the matrix. When *Mathematica* suspects that you are trying to invert a singular numerical matrix, it prints a warning.

Mathematica prints a warning if you invert a numerical matrix that it suspects is singular.

```
In[13]:= Inverse[ {{1., 2.}, {1., 2.}} ]
Inverse::cond:
      Warning: Inverse encountered badly conditioned
         matrix {{1., 2.}, {1., 2.}}
       , result may be meaningless.

                      33             33
Out[13]= {{1.42972 10  , -1.42972 10  },

                 32            32
    {-7.14862 10  , 7.14862 10  }}
```

If you work with high precision approximate numbers, *Mathematica* will keep track of the precision of matrix inverses that you generate.

This generates a 6×6 numerical matrix with entries of 50-digit precision.

```
In[14]:= m = N [ Table[ Exp[i j], {i, 6}, {j, 6} ], 50 ] ;
```

This takes the matrix, multiplies it by its inverse, and shows the first row of the result. The off-diagonal terms are zero to a precision of nearly 50 digits.

```
In[15]:= (m . Inverse[m]) [[1]]

                          -53         -57          -58
Out[15]= {1., 0., -1. 10     , 1. 10     , -2. 10     ,

                -61
          -1. 10   }
```

This generates a 50-digit numerical approximation to a 6×6 Hilbert matrix. Hilbert matrices are notoriously hard to invert numerically.

```
In[16]:= m = N[Table[1/(i + j - 1), {i, 6}, {j, 6}], 50] ;
```

In this case, the off-diagonal terms are zero to an accuracy of less than 50 digits. The larger off-diagonal terms are a reflection of the "bad numerical conditioning" of this matrix.

```
In[17]:= (m . Inverse[m]) [[1]]

                          -39                    -39
Out[17]= {1., 0., -3. 10     , 0., 0., -3. 10     }
```

`Inverse` works only on square matrices. Section 3.7.11 discusses the function `PseudoInverse`, which can also be used with non-square matrices.

Web sample page from The *Mathematica* Book, First Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-19334-5; softcover ISBN 0-201-19330-2). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.