A.4 Simplification and Transformation Rules

A.4.1 Simplification Rules for Combinatorial Functions

Mathematica has a number of "simplification rules" for combinatorial functions built in. You can, however, add further rules. Notice that you have to call `Unprotect` before making definitions for built-in objects.

```
Unprotect[Factorial, Binomial]

Factorial/: (k_)! k1_ := (k1)! /; k1 - k == 1
Factorial/: (n_)!/(m_)! := Product[i, {i, m+1, n}] /; n - m > 0 && IntegerQ[n-m]
Factorial/: (n_)!/(m_)! := 1/Product[i, {i, n+1, m}] /; m - n > 0 && IntegerQ[m-n]

Binomial/: Binomial[n_, k_]/Binomial[n_, k1_] := (n-k+1)/k /; k1 == k-1
Binomial/: Binomial[n_, k1_]/Binomial[n_, k_] := k/(n-k+1) /; k1 == k-1
```

Simplification rules for combinatorial functions.

Mathematica does not automatically simplify this expression.

```
In[1]:= (n + 2)! / n!

        (2 + n)!
Out[1]= --------
           n!
```

This reads in the package of combinatorial simplification rules.

```
In[2]:= <<CombinatorialSimplification.m
```

Now *Mathematica* does simplify the expression.

```
In[3]:= (n + 2)! / n!
Out[3]= (1 + n) (2 + n)
```

Web sample page from The *Mathematica* Book, First Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-19334-5; softcover ISBN 0-201-19330-2). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.