■ 1.8.12 Plotting Lists of Data

So far, we have discussed how you can use *Mathematica* to make plots of *functions.* You give *Mathematica* a function, and it builds up a curve or surface by evaluating the function at many different points.

This section describes how you can make plots from lists of data, instead of functions. The *Mathematica* commands for plotting lists of data are direct analogs of the ones discussed above for plotting functions.

`ListPlot[{`y_1`, `y_2`, ...}]`	plot y_1, y_2, ... at x values 1, 2, ...
`ListPlot[{{`x_1`, `y_1`}, {`x_2`, `y_2`}, ...}]`	plot points (x_1, y_1), ...
`ListPlot[`*list*`, PlotJoined -> True]`	plot a line through the points
`ListPlot3D[{{`z_{11}`, `z_{12}`, ...}, {`z_{21}`, `z_{22}`, ...}, ...}]`	make a three-dimensional plot of the array of heights z_{xy}
`ListContourPlot[`*array*`]`	make a contour plot from an array of heights
`ListDensityPlot[`*array*`]`	make a density plot

Functions for plotting lists of data.

Here is a list of values.

```
In[1]:= t = Table[i^2, {i, 10}]
Out[1]= {1, 4, 9, 16, 25, 36, 49, 64, 81, 100}
```

This plots the values.

```
In[2]:= ListPlot[t]
```



Web sample page from The *Mathematica* Book, First Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-19334-5; softcover ISBN 0-201-19330-2). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.

This puts a line through the points.

```
In[3]:= ListPlot[t, PlotJoined -> True]
```



This gives a list of x, y pairs.

```
In[4]:= Table[{i^2, 4 i^2 + i^3}, {i, 10}]

Out[4]= {{1, 5}, {4, 24}, {9, 63}, {16, 128}, {25, 225},
     {36, 360}, {49, 539}, {64, 768}, {81, 1053},
     {100, 1400}}
```

This plots the points.

```
In[5]:= ListPlot[%]
```



This gives a rectangular array of values. The array is quite large, so we end the input with a semicolon to stop the result from being printed out.

```
In[6]:= t3 = Table[Mod[y, x], {x, 20}, {y, 20}] ;
```

This makes a three-dimensional plot of the array of values.

```
In[7]:= ListPlot3D[t3]
```



You can redraw the plot using Show, as usual.

```
In[8]:= Show[%, ViewPoint -> {1.5, -0.5, 0}]
```



This gives a density plot of the array of values.

```
In[9]:= ListDensityPlot[t3]
```