■ 2.2.6 A First Encounter with Patterns

The `x_` that appears in `f[x_]` is an example of an important kind of object in *Mathematica* known as a *pattern*. Patterns are a way of representing *classes of expressions* in *Mathematica*. A single pattern can stand for a whole class of possible expressions. By giving transformation rules for patterns, you can define rules for classes of expressions.

The basic object that appears in *Mathematica* patterns is `_` (pronounced "blank"). The fundamental rule is that `_` *stands for any expression*. As an example, the pattern `f[_]` can thus stand for any expression of the form `f[`*anything*`]`. The pattern `x^_` could stand for any power of `x`. The pattern `_^_` represents any power of any expression.

In general, a pattern can stand for a particular expression if there is a way of "filling in the blanks" (`_` objects) in the pattern so as to get the expression. `_^_` stands for `(1 + x)^3` because the first `_` can be filled in with `1 + x` and the second one with `3`.

Each time a particular pattern is used, say in a transformation rule, `_` can stand for a different expression. In one case, `_` might stand for `a + b`; in another `2 - a`. It is often convenient to name the expression that `_` stands for in a particular case.

The pattern *x*`_` can again stand for any expression, but now the expression is given the *name* *x*. You can use the symbol *x* to represent the expression, say on the right-hand side of a definition.

`f[n_]`	`f` with any "argument", named `n`
`f[n_, m_]`	`f` with two arguments, named `n` and `m`
`x^n_`	`x` to any power, with the power named `n`
`x_^n_`	any expression to any power
`a_ + b_`	a sum of two expressions
`{a1_, a2_}`	a list of two expressions
`f[n_, n_]`	`f` with two *identical* arguments

Some examples of patterns.

This defines a rule for `g` when its argument is a power.

```
In[1]:= g[x_^y_] :=  y h[x]
```

Web sample page from The *Mathematica* Book, First Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-19334-5; softcover ISBN 0-201-19330-2). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.

This uses the rule for g.

```
In[2]:= g[(1 + x)^2]
Out[2]= 2 h[1 + x]
```

Here is a more complicated transformation rule.

```
In[3]:= g[(x_ + y_)^n_ + c_] := n hp[x, y] + c
```

This uses the more complicated transformation rule.

```
In[4]:= g[(a + b)^3 + x^2]
                2
Out[4]= x  + 3 hp[a, b]
```

When you have a pattern in which a named blank such as `n_` occurs several times, the blank must stand for the *same expression* each time. As a result, a pattern like `f[n_,n_]` can stand for `f[a+b, a+b]`, but not for `f[a, b]`.

This defines a value for `f` with two identical arguments.

```
In[5]:= f[n_, n_] := 2n
```

This defines a value for the general case of two arbitrary arguments.

```
In[6]:= f[n_, m_] := g[n+m]
```

This shows the values defined for `f`. Notice that the more specific case with two identical arguments is given first.

```
In[7]:= ?f
f
f/: f[n_, n_] := 2 n
f/: f[n_, m_] := g[n + m]
```

With two identical arguments, the first rule for `f` is used.

```
In[7]:= f[1 + x, 1 + x]
Out[7]= 2 (1 + x)
```

With different arguments, the second rule is used.

```
In[8]:= f[1 - x, 1 + x]
Out[8]= g[2]
```

Web sample page from The *Mathematica* Book, First Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-19334-5; softcover ISBN 0-201-19330-2). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.