A.4.2 Trigonometric Simplification Rules

If you make assignments that implement simplification rules, then these rules will be used whenever they apply. Often, however, you want to set up rules that can be used in a more controlled fashion.

This section shows part of a package for reducing trigonometric expressions, using a sequence of transformation rules. The goal is to write expressions that involve trigonometric functions of multiple angles in terms of powers of trigonometric functions.

The package defines some lists of transformation rules, then applies these rules using the `/.` and `//.` operators.

```
TrigReduceRel = {
    Cos[(n_Integer?Positive) x_] :> 2^(n-1) Cos[x]^n +
        Sum[ Binomial[n-i-1, i-1] (-1)^i n/i 2^(n-2i-1) Cos[x]^(n-2i),
           {i, 1, n/2} ],

    Sin[(m_Integer?OddQ) x_] :> Block[{p = -(m^2-1)/6, s = Sin[x]},
        Do[s += p Sin[x]^k;
           p = p * -(m^2 - k^2)/(k+2)/(k+1),
          {k, 3, m, 2}];
        m s]                            /; m > 0,

    Sin[(n_Integer?EvenQ) x_] :>
        Sum[ Binomial[n, i] (-1)^((i-1)/2) Sin[x]^i Cos[x]^(n-i),
             {i, 1, n, 2} ]             /; n > 0,

    Sin[x_ + y_] :> Sin[x] Cos[y] + Sin[y] Cos[x],
    Sin[x_ - y_] :> Sin[x] Cos[y] - Sin[y] Cos[x],
    Cos[x_ + y_] :> Cos[x] Cos[y] - Sin[x] Sin[y],
    Cos[x_ - y_] :> Cos[x] Cos[y] + Sin[x] Sin[y],
}

TrigCanonicalRel = {
    (* Sin is an odd function *)
    Sin[(n_?Negative) x_.]     :> -Sin[-n x],
    Sin[(n_?Negative) x_ + y_] :> -Sin[-n x - y] /; Order[x, y] == 1,

    (* Cos is an even function *)
    Cos[(n_?Negative) x_.]     :>  Cos[-n x] /; n < 0,
    Cos[(n_?Negative) x_ + y_] :>  Cos[-n x - y] /; Order[x, y] == 1,

    a_. Sin[x_]^2 + a_. Cos[x_]^2 :> a,
    n_Integer a_. Sin[x_]^2 + m_Integer a_. Cos[x_]^2 :> a (m + (n-m)Sin[x]^2)
}
```

Web sample page from The *Mathematica* Book, First Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-19334-5; softcover ISBN 0-201-19330-2). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.

```
TrigReduce[e_] :=
        e /. TrigCanonicalRel //. TrigReduceRel /. TrigCanonicalRel
```

Trigonometric simplification rules.

Read in the trigonometric simplification package.

```
In[1]:= <<Trigonometry.m
Out[1]= Trigonometry`
```

Mathematica makes no automatic transformations on this expression.

```
In[2]:= Sin[4x]
Out[2]= Sin[4 x]
```

`TrigReduce` converts the trigonometric function of a multiple angle into powers of trigonometric functions of single angles.

```
In[3]:= TrigReduce[%]
                   3                        3
Out[3]= 4 Cos[x]  Sin[x] - 4 Cos[x] Sin[x]
```

Here is a slightly more complicated case.

```
In[4]:= TrigReduce[Sin[2x + y]]
                                                       2
Out[4]= 2 Cos[x] Cos[y] Sin[x] + (-1 + 2 Cos[x] ) Sin[y]
```