2.2.7 Specifying Mathematical Relations

One of the most important ways you can use transformation rules for patterns is to specify *mathematical relations*. If you look through a handbook of mathematics, you will find many formulae that give relations between mathematical functions.

One such mathematical relation might be $\log(\exp(x)) = x$.

You can interpret this relation as being a transformation rule for a pattern. It states that whenever you have an expression of the form `log[exp[`*anything*`]]`, it can be "simplified" to *anything*.

You can enter this transformation rule almost directly into *Mathematica*, just by giving the definition `log[exp[x_]] := x`.

This specifies a relation for the logarithm function.

```
In[1]:= log[exp[x_]] := x
```

Now the relation is used to "simplify" expressions involving `log`.

```
In[2]:= log[exp[y^2+1]]

             2
Out[2]= 1 + y
```

You can now go on and add other transformation rules for the `log` function.

Two standard mathematical relations for logarithm functions are: $\log(xy) = \log(x) + \log(y)$; $\log(x^n) = n\ \log(x)$.

Once again, these relations can be entered almost directly into *Mathematica* as transformation rules for patterns: `log[x_ y_] := log[x] + log[y]`; `log[x_^n_] := n log[x]`.

This gives the rule for expanding logarithms of products.

```
In[3]:= log[x_ y_] := log[x] + log[y]
```

Here is the rule for logarithms of powers.

```
In[4]:= log[x_^n_] := n log[x]
```

This shows all the rules defined for `log`.

```
In[5]:= ?log

log
log/: log[exp[x_]] := x

log/: log[(x_) (y_)] := log[x] + log[y]

              n_
log/: log[(x_)  ] := n log[x]
```

This is "simplified" using the rule for logarithms of powers.

```
In[5]:= log[(1+x)^5]

Out[5]= 5 log[1 + x]
```

Web sample page from The *Mathematica* Book, First Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-19334-5; softcover ISBN 0-201-19330-2). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.

This is simplified using the rule for logarithms of products. The rule gives the expansion for a product of just two expressions, so it had to be applied three times in this case.

```
In[6]:= log[a b c d]

Out[6]= log[a] + log[b] + log[c] + log[d]
```

Here is a more complicated example. The various transformation rules you have defined for `log` are each used several times to get the final result.

```
In[7]:= log[(1+x)^2 y^2] + log[x^3 (1-y)^4] +
               log[exp[a b] x^2]

Out[7]= a b + 5 log[x] + 2 log[y] + 2 log[1 + x] +

     4 log[1 - y]
```

These examples begin to show the power of "mathematical programming" in *Mathematica*. Section 2.3 gives some more details of how to use patterns in specifying mathematical relations. Section 4.1 discusses the general methodology of "mathematical programming".