3.1.3 Output Forms for Numbers

`NumberForm[`*expr*`, `*n*`]`	print real numbers in *expr* to *n* digit precision
`ScientificForm[`*expr*`]`	print all real numbers in scientific notation
`EngineeringForm[`*expr*`]`	print all real numbers in engineering notation (exponents divisible by 3)

Output forms for numbers in *Mathematica*.

Here is π^{60} to 35 decimal places.

```
In[1]:= x = N[Pi^60, 35]
Out[1]= 674516164233396768200126958915.55914
```

This prints the number to 10 places.

```
In[2]:= NumberForm[x, 10]
                       29
Out[2]= 6.745161642 10
```

`EngineeringForm` writes real numbers so that their exponents are divisible by three.

```
In[3]:= EngineeringForm[x]
                                                 27
Out[3]= 674.51616423339676820012695891555914 10
```

You can give options to `NumberForm` to specify in more detail how you want both real numbers and integers to be printed.

option	*default value*	
`DigitBlock`	`Infinity`	number of digits between breaks
`NumberSeparator`	`","`	string to insert at breaks between blocks of digits
`NumberPoint`	`"."`	decimal point string
`ExponentStep`	`1`	size of steps in exponent

Options for `NumberForm`.

Setting `DigitBlock->`*n* makes `NumberForm` insert a separator every *n* digits.

```
In[4]:= NumberForm[30!, DigitBlock->3]
Out[4]= 265,252,859,812,191,058,636,308,480,000,000
```

This writes the number in blocks of five digits, with spaces in between.

```
In[5]:= NumberForm[30!, DigitBlock->5,
                    NumberSeparator->" "]

Out[5]= 265 25285 98121 91058 63630 84800 00000
```